


A N A L O G P O W E R

Analog Power Inc.

SPICE MODELING Revised August 2015





Analog Power Spice Model Goals

- To provide simple, fast, accurate SPICE model to be used predominantly for switching speed and loss calculations
- To be able to support model and its use
- To allow temperature modeling, but not at the expense of speed and simplicity
- To focus on the useful modeling parameters, and do not include breakdown voltage etc as stresses can be checked via model probing
- To support on low cost universal platforms, WinSpice, NGSPICE, LT Spice
- To be able to use SPICE to select the optimum part for PWM applications


Not Intended for (significant added complexity/run time):

- Modeling of breakdown voltage, high temperature leakage
- Thermal modeling including RDS variation with temperature
 - Model can be adjusted for temperature
- Qrr modeling

Analog Power SPICE Subcircuit

Analog Power model R1.0:


- Simple Level 1 model with 8 external components
- Excellent fitting to data sheet curves
- All parameters are device-derived not process derived
- Very simple to see effects of each parameter



Analog Power SPICE Subcircuit 2.0

Improved implementation:
Non-Linear dependent voltage sources used to set dv/dt over diode and fixed capacitor = zero appropriately , and therefore isolate them when not in circuit.

No switches used, but identical concept



Analog Power Spice Example (switch, L1)

```
.SUBCKT AM90N08-04B 1 2 3
*Nom Temp=25 deg C
Mos1 4 5 6 6 APLMOS w=1 l=1
.MODEL APLMOS NMOS (Level=1 VTO=1.94 KP=116 CGDO = 600p CGSO = 9100p)
Rdrain 1 4 4.7e-3
Dbody 6 1 BodyDiode
.MODEL BodyDiode D (RS=2e-3 IS=2e-12 vj=0.6 RS=2m CJO=1000p M=0.67)
Lsource 3 6 3e-9
Rgate 2 5 1.8
Dcgd 5 7 Crss
.MODEL Crss D (RS=1e-3 CJO=10.9n M=1.18)
Sdiode 7 1 1 5 Sfb1
Scap 8 1 5 1 Sfb2
.MODEL Sfb1 VSWITCH (RON=100m ROFF=10e11 VON=0 VOFF=-0.01)
.MODEL Sfb2 VSWITCH (RON=100m ROFF=10e11 VON=0.01 VOFF=0)
C2 8 5 10.9n IC=0
R2 8 5 1e8
.ENDS
```

NOTE some Spice versions (E.G. NGSPICE) use different syntax for the switches:

```
Sdiode 7 1 1 5 Sfb1 ON
Scap 8 1 5 1 Sfb2 Off
.MODEL Sfb1 SW (RON=0.1 ROFF=10e11 VT=0 VH=-0.01)
.MODEL Sfb2 SW (RON=0.1 ROFF=10e11 VT=0.01 VH=0)
```

Analog Power Spice Example (no switch, L3)

```

* Analog Power Spice Level 3 Model - fixed temperature
* Rev 4/14/2014 PD
* Level 3 model is experimental, sub-threshold leakage is overestimated, which will affect VSD diode
  modeling
.SUBCKT AM3446N 1 2 3
*Nom Temp=25 deg C
Mos1 4 5 6 6 APLMOS l=0.25u w=1
.MODEL APLMOS NMOS (LEVEL = 3 vto=0.91 KP =10u NSUB=2e+17 Kappa = 0.12 CGDO = 55p
  CGSO = 336p)
Rdrain 1 4 28m
Dbody 6 1 BodyD
.MODEL BodyD D (IS=0.5e-12 RS=30e-3 CJO=10p M=0.2)
Lsource 3 6 0.3e-9
Rgate 2 5 3
Dcgd 5 7 Crss
.MODEL Crss D (RS=1e-3 CJO=400p M=0.91)
B1 7 1 V= (abs(v(5)-V(1))+v(5)-V(1))/2
C2 8 5 400p IC=0
R2 8 5 1e9
B2 1 8 V= (abs(v(1)-V(5))+v(1)-V(5))/2
.ENDS
  
```

Level 1 and Level 3 Models



- Many vendors add many external components and use more complicated BSIM models, but do they work any better?
- Parameter-derived level 1 model is recognized as suitable for discrete devices for modeling switching time etc
 - more complicated models are intended for use in IC modeling and are process-derived
- However level 1 model does **not** model RDS variation with VGS well for modern devices and switching characteristics may suffer
 - Use Analog Power Level 3 model instead
- Analog Power models are device derived, as opposed to process and theory derived and therefore more accurate
- Rather than include temperature dependencies, our simple model can be accurately tweaked for any temperature and this provides a faster iteration to modeling at higher temperatures

Transfer Characteristics

- Level 1 model uses simple equation:

$$I_D \propto (V_{GS} - V_{GS(TH)})^2$$


- This model holds true for state of the art Trench MOSFETs, especially surface mount where current density is quite low
- KP (Spice model parameter) easy to derive from characterization data. L1 model gives accurate transfer function, **but inaccurate RDS VS VGS plot**
- Level 3 model allows KP to reflect accurate RDS VDS VGS yet still provides accurate Transfer characteristics




CRSS – The key to accuracy

Three components:


- CGDO – part of level 1 model
 - does not vary with VDS
- Reverse biased diode
 - gives variation with VDS
 - removed when $V_{GD} > 0$
- Fixed CDG2 capacitor
 - switched in when $V_{GD} > 0$
 - 1.2 X value of diode at $V_{DG} = 0$ to reflect higher CRSS for +ve V_{GD}
- The combination of a diode with fixed parallel capacitance gives fit with data
- Alternate implementation with voltage sources works in identical manner, but use of switches is avoided – simplifies syntax, and avoids SPICE transients



Comparison Data Sheet Vs. Model



Data sheet



Model



Conclusion

Analog Power proposed SPICE subcircuit:

- Simple and therefore fast, no switches, no extra active components
- Device derived. Accurate in operation (switching, RDS)
- L3: Developed predominantly for optimum part selection in PWM applications
 - L1 most accurate for linear regulator type applications
- Device-derived and therefore contains accurate parameters
- Easily adjusted to any operating temperature
- Allows what-if experiments, parameters are easily identifiable and editable
- L3 model includes accurate transfer AND RDS Vs VGS modeling